(*  Title:      ZF/Cardinal.thy
    Author:     Lawrence C Paulson, Cambridge University Computer Laboratory
    Copyright   1994  University of Cambridge
*)

section\<open>Cardinal Numbers Without the Axiom of Choice\<close>

theory Cardinal imports OrderType Finite Nat_ZF Sum begin

definition
  (*least ordinal operator*)
   Least    :: "(i=>o) => i"    (binder "\<mu> " 10)  where
     "Least(P) == THE i. Ord(i) & P(i) & (\<forall>j. j<i \<longrightarrow> ~P(j))"

definition
  eqpoll   :: "[i,i] => o"     (infixl "\<approx>" 50)  where
    "A \<approx> B == \<exists>f. f \<in> bij(A,B)"

definition
  lepoll   :: "[i,i] => o"     (infixl "\<lesssim>" 50)  where
    "A \<lesssim> B == \<exists>f. f \<in> inj(A,B)"

definition
  lesspoll :: "[i,i] => o"     (infixl "\<prec>" 50)  where
    "A \<prec> B == A \<lesssim> B & ~(A \<approx> B)"

definition
  cardinal :: "i=>i"           ("|_|")  where
    "|A| == (\<mu> i. i \<approx> A)"

definition
  Finite   :: "i=>o"  where
    "Finite(A) == \<exists>n\<in>nat. A \<approx> n"

definition
  Card     :: "i=>o"  where
    "Card(i) == (i = |i|)"

subsection\<open>The Schroeder-Bernstein Theorem\<close>
text\<open>See Davey and Priestly, page 106\<close>

(** Lemma: Banach's Decomposition Theorem **)

lemma decomp_bnd_mono: "bnd_mono(X, %W. X - g``(Y - f``W))"
by (rule bnd_monoI, blast+)

lemma Banach_last_equation:
    "g \<in> Y->X
     ==> g``(Y - f`` lfp(X, %W. X - g``(Y - f``W))) =
         X - lfp(X, %W. X - g``(Y - f``W))"
apply (rule_tac P = "%u. v = X-u" for v
       in decomp_bnd_mono [THEN lfp_unfold, THEN ssubst])
apply (simp add: double_complement  fun_is_rel [THEN image_subset])
done

lemma decomposition:
     "[| f \<in> X->Y;  g \<in> Y->X |] ==>
      \<exists>XA XB YA YB. (XA \<inter> XB = 0) & (XA \<union> XB = X) &
                      (YA \<inter> YB = 0) & (YA \<union> YB = Y) &
                      f``XA=YA & g``YB=XB"
apply (intro exI conjI)
apply (rule_tac [6] Banach_last_equation)
apply (rule_tac [5] refl)
apply (assumption |
       rule  Diff_disjoint Diff_partition fun_is_rel image_subset lfp_subset)+
done

lemma schroeder_bernstein:
    "[| f \<in> inj(X,Y);  g \<in> inj(Y,X) |] ==> \<exists>h. h \<in> bij(X,Y)"
apply (insert decomposition [of f X Y g])
apply (simp add: inj_is_fun)
apply (blast intro!: restrict_bij bij_disjoint_Un intro: bij_converse_bij)
(* The instantiation of exI to @{term"restrict(f,XA) \<union> converse(restrict(g,YB))"}
   is forced by the context!! *)
done

(** Equipollence is an equivalence relation **)

lemma bij_imp_eqpoll: "f \<in> bij(A,B) ==> A \<approx> B"
apply (unfold eqpoll_def)
apply (erule exI)
done

(*A \<approx> A*)
lemmas eqpoll_refl = id_bij [THEN bij_imp_eqpoll, simp]

lemma eqpoll_sym: "X \<approx> Y ==> Y \<approx> X"
apply (unfold eqpoll_def)
apply (blast intro: bij_converse_bij)
done

lemma eqpoll_trans [trans]:
    "[| X \<approx> Y;  Y \<approx> Z |] ==> X \<approx> Z"
apply (unfold eqpoll_def)
apply (blast intro: comp_bij)
done

(** Le-pollence is a partial ordering **)

lemma subset_imp_lepoll: "X<=Y ==> X \<lesssim> Y"
apply (unfold lepoll_def)
apply (rule exI)
apply (erule id_subset_inj)
done

lemmas lepoll_refl = subset_refl [THEN subset_imp_lepoll, simp]

lemmas le_imp_lepoll = le_imp_subset [THEN subset_imp_lepoll]

lemma eqpoll_imp_lepoll: "X \<approx> Y ==> X \<lesssim> Y"
by (unfold eqpoll_def bij_def lepoll_def, blast)

lemma lepoll_trans [trans]: "[| X \<lesssim> Y;  Y \<lesssim> Z |] ==> X \<lesssim> Z"
apply (unfold lepoll_def)
apply (blast intro: comp_inj)
done

lemma eq_lepoll_trans [trans]: "[| X \<approx> Y;  Y \<lesssim> Z |] ==> X \<lesssim> Z"
 by (blast intro: eqpoll_imp_lepoll lepoll_trans)

lemma lepoll_eq_trans [trans]: "[| X \<lesssim> Y;  Y \<approx> Z |] ==> X \<lesssim> Z"
 by (blast intro: eqpoll_imp_lepoll lepoll_trans)

(*Asymmetry law*)
lemma eqpollI: "[| X \<lesssim> Y;  Y \<lesssim> X |] ==> X \<approx> Y"
apply (unfold lepoll_def eqpoll_def)
apply (elim exE)
apply (rule schroeder_bernstein, assumption+)
done

lemma eqpollE:
    "[| X \<approx> Y; [| X \<lesssim> Y; Y \<lesssim> X |] ==> P |] ==> P"
by (blast intro: eqpoll_imp_lepoll eqpoll_sym)

lemma eqpoll_iff: "X \<approx> Y \<longleftrightarrow> X \<lesssim> Y & Y \<lesssim> X"
by (blast intro: eqpollI elim!: eqpollE)

lemma lepoll_0_is_0: "A \<lesssim> 0 ==> A = 0"
apply (unfold lepoll_def inj_def)
apply (blast dest: apply_type)
done

(*@{term"0 \<lesssim> Y"}*)
lemmas empty_lepollI = empty_subsetI [THEN subset_imp_lepoll]

lemma lepoll_0_iff: "A \<lesssim> 0 \<longleftrightarrow> A=0"
by (blast intro: lepoll_0_is_0 lepoll_refl)

lemma Un_lepoll_Un:
    "[| A \<lesssim> B; C \<lesssim> D; B \<inter> D = 0 |] ==> A \<union> C \<lesssim> B \<union> D"
apply (unfold lepoll_def)
apply (blast intro: inj_disjoint_Un)
done

(*A \<approx> 0 ==> A=0*)
lemmas eqpoll_0_is_0 = eqpoll_imp_lepoll [THEN lepoll_0_is_0]

lemma eqpoll_0_iff: "A \<approx> 0 \<longleftrightarrow> A=0"
by (blast intro: eqpoll_0_is_0 eqpoll_refl)

lemma eqpoll_disjoint_Un:
    "[| A \<approx> B;  C \<approx> D;  A \<inter> C = 0;  B \<inter> D = 0 |]
     ==> A \<union> C \<approx> B \<union> D"
apply (unfold eqpoll_def)
apply (blast intro: bij_disjoint_Un)
done

subsection\<open>lesspoll: contributions by Krzysztof Grabczewski\<close>

lemma lesspoll_not_refl: "~ (i \<prec> i)"
by (simp add: lesspoll_def)

lemma lesspoll_irrefl [elim!]: "i \<prec> i ==> P"
by (simp add: lesspoll_def)

lemma lesspoll_imp_lepoll: "A \<prec> B ==> A \<lesssim> B"
by (unfold lesspoll_def, blast)

lemma lepoll_well_ord: "[| A \<lesssim> B; well_ord(B,r) |] ==> \<exists>s. well_ord(A,s)"
apply (unfold lepoll_def)
apply (blast intro: well_ord_rvimage)
done

lemma lepoll_iff_leqpoll: "A \<lesssim> B \<longleftrightarrow> A \<prec> B | A \<approx> B"
apply (unfold lesspoll_def)
apply (blast intro!: eqpollI elim!: eqpollE)
done

lemma inj_not_surj_succ:
  assumes fi: "f \<in> inj(A, succ(m))" and fns: "f \<notin> surj(A, succ(m))"
  shows "\<exists>f. f \<in> inj(A,m)"
proof -
  from fi [THEN inj_is_fun] fns
  obtain y where y: "y \<in> succ(m)" "\<And>x. x\<in>A \<Longrightarrow> f ` x \<noteq> y"
    by (auto simp add: surj_def)
  show ?thesis
    proof
      show "(\<lambda>z\<in>A. if f`z = m then y else f`z) \<in> inj(A, m)" using y fi
        by (simp add: inj_def)
           (auto intro!: if_type [THEN lam_type] intro: Pi_type dest: apply_funtype)
      qed
qed

(** Variations on transitivity **)

lemma lesspoll_trans [trans]:
      "[| X \<prec> Y; Y \<prec> Z |] ==> X \<prec> Z"
apply (unfold lesspoll_def)
apply (blast elim!: eqpollE intro: eqpollI lepoll_trans)
done

lemma lesspoll_trans1 [trans]:
      "[| X \<lesssim> Y; Y \<prec> Z |] ==> X \<prec> Z"
apply (unfold lesspoll_def)
apply (blast elim!: eqpollE intro: eqpollI lepoll_trans)
done

lemma lesspoll_trans2 [trans]:
      "[| X \<prec> Y; Y \<lesssim> Z |] ==> X \<prec> Z"
apply (unfold lesspoll_def)
apply (blast elim!: eqpollE intro: eqpollI lepoll_trans)
done

lemma eq_lesspoll_trans [trans]:
      "[| X \<approx> Y; Y \<prec> Z |] ==> X \<prec> Z"
  by (blast intro: eqpoll_imp_lepoll lesspoll_trans1)

lemma lesspoll_eq_trans [trans]:
      "[| X \<prec> Y; Y \<approx> Z |] ==> X \<prec> Z"
  by (blast intro: eqpoll_imp_lepoll lesspoll_trans2)

(** \<mu> -- the least number operator [from HOL/Univ.ML] **)

lemma Least_equality:
    "[| P(i);  Ord(i);  !!x. x<i ==> ~P(x) |] ==> (\<mu> x. P(x)) = i"
apply (unfold Least_def)
apply (rule the_equality, blast)
apply (elim conjE)
apply (erule Ord_linear_lt, assumption, blast+)
done

lemma LeastI:
  assumes P: "P(i)" and i: "Ord(i)" shows "P(\<mu> x. P(x))"
proof -
  { from i have "P(i) \<Longrightarrow> P(\<mu> x. P(x))"
      proof (induct i rule: trans_induct)
        case (step i)
        show ?case
          proof (cases "P(\<mu> a. P(a))")
            case True thus ?thesis .
          next
            case False
            hence "\<And>x. x \<in> i \<Longrightarrow> ~P(x)" using step
              by blast
            hence "(\<mu> a. P(a)) = i" using step
              by (blast intro: Least_equality ltD)
            thus ?thesis using step.prems
              by simp
          qed
      qed
  }
  thus ?thesis using P .
qed

text\<open>The proof is almost identical to the one above!\<close>
lemma Least_le:
  assumes P: "P(i)" and i: "Ord(i)" shows "(\<mu> x. P(x)) \<le> i"
proof -
  { from i have "P(i) \<Longrightarrow> (\<mu> x. P(x)) \<le> i"
      proof (induct i rule: trans_induct)
        case (step i)
        show ?case
          proof (cases "(\<mu> a. P(a)) \<le> i")
            case True thus ?thesis .
          next
            case False
            hence "\<And>x. x \<in> i \<Longrightarrow> ~ (\<mu> a. P(a)) \<le> i" using step
              by blast
            hence "(\<mu> a. P(a)) = i" using step
              by (blast elim: ltE intro: ltI Least_equality lt_trans1)
            thus ?thesis using step
              by simp
          qed
      qed
  }
  thus ?thesis using P .
qed

(*\<mu> really is the smallest*)
lemma less_LeastE: "[| P(i);  i < (\<mu> x. P(x)) |] ==> Q"
apply (rule Least_le [THEN [2] lt_trans2, THEN lt_irrefl], assumption+)
apply (simp add: lt_Ord)
done

(*Easier to apply than LeastI: conclusion has only one occurrence of P*)
lemma LeastI2:
    "[| P(i);  Ord(i);  !!j. P(j) ==> Q(j) |] ==> Q(\<mu> j. P(j))"
by (blast intro: LeastI )

(*If there is no such P then \<mu> is vacuously 0*)
lemma Least_0:
    "[| ~ (\<exists>i. Ord(i) & P(i)) |] ==> (\<mu> x. P(x)) = 0"
apply (unfold Least_def)
apply (rule the_0, blast)
done

lemma Ord_Least [intro,simp,TC]: "Ord(\<mu> x. P(x))"
proof (cases "\<exists>i. Ord(i) & P(i)")
  case True
  then obtain i where "P(i)" "Ord(i)"  by auto
  hence " (\<mu> x. P(x)) \<le> i"  by (rule Least_le)
  thus ?thesis
    by (elim ltE)
next
  case False
  hence "(\<mu> x. P(x)) = 0"  by (rule Least_0)
  thus ?thesis
    by auto
qed

subsection\<open>Basic Properties of Cardinals\<close>

(*Not needed for simplification, but helpful below*)
lemma Least_cong: "(!!y. P(y) \<longleftrightarrow> Q(y)) ==> (\<mu> x. P(x)) = (\<mu> x. Q(x))"
by simp

(*Need AC to get @{term"X \<lesssim> Y ==> |X| \<le> |Y|"};  see well_ord_lepoll_imp_Card_le
  Converse also requires AC, but see well_ord_cardinal_eqE*)
lemma cardinal_cong: "X \<approx> Y ==> |X| = |Y|"
apply (unfold eqpoll_def cardinal_def)
apply (rule Least_cong)
apply (blast intro: comp_bij bij_converse_bij)
done

(*Under AC, the premise becomes trivial; one consequence is ||A|| = |A|*)
lemma well_ord_cardinal_eqpoll:
  assumes r: "well_ord(A,r)" shows "|A| \<approx> A"
proof (unfold cardinal_def)
  show "(\<mu> i. i \<approx> A) \<approx> A"
    by (best intro: LeastI Ord_ordertype ordermap_bij bij_converse_bij bij_imp_eqpoll r)
qed

(* @{term"Ord(A) ==> |A| \<approx> A"} *)
lemmas Ord_cardinal_eqpoll = well_ord_Memrel [THEN well_ord_cardinal_eqpoll]

lemma Ord_cardinal_idem: "Ord(A) \<Longrightarrow> ||A|| = |A|"
 by (rule Ord_cardinal_eqpoll [THEN cardinal_cong])

lemma well_ord_cardinal_eqE:
  assumes woX: "well_ord(X,r)" and woY: "well_ord(Y,s)" and eq: "|X| = |Y|"
shows "X \<approx> Y"
proof -
  have "X \<approx> |X|" by (blast intro: well_ord_cardinal_eqpoll [OF woX] eqpoll_sym)
  also have "... = |Y|" by (rule eq)
  also have "... \<approx> Y" by (rule well_ord_cardinal_eqpoll [OF woY])
  finally show ?thesis .
qed

lemma well_ord_cardinal_eqpoll_iff:
     "[| well_ord(X,r);  well_ord(Y,s) |] ==> |X| = |Y| \<longleftrightarrow> X \<approx> Y"
by (blast intro: cardinal_cong well_ord_cardinal_eqE)

(** Observations from Kunen, page 28 **)

lemma Ord_cardinal_le: "Ord(i) ==> |i| \<le> i"
apply (unfold cardinal_def)
apply (erule eqpoll_refl [THEN Least_le])
done

lemma Card_cardinal_eq: "Card(K) ==> |K| = K"
apply (unfold Card_def)
apply (erule sym)
done

(* Could replace the  @{term"~(j \<approx> i)"}  by  @{term"~(i \<preceq> j)"}. *)
lemma CardI: "[| Ord(i);  !!j. j<i ==> ~(j \<approx> i) |] ==> Card(i)"
apply (unfold Card_def cardinal_def)
apply (subst Least_equality)
apply (blast intro: eqpoll_refl)+
done

lemma Card_is_Ord: "Card(i) ==> Ord(i)"
apply (unfold Card_def cardinal_def)
apply (erule ssubst)
apply (rule Ord_Least)
done

lemma Card_cardinal_le: "Card(K) ==> K \<le> |K|"
apply (simp (no_asm_simp) add: Card_is_Ord Card_cardinal_eq)
done

lemma Ord_cardinal [simp,intro!]: "Ord(|A|)"
apply (unfold cardinal_def)
apply (rule Ord_Least)
done

text\<open>The cardinals are the initial ordinals.\<close>
lemma Card_iff_initial: "Card(K) \<longleftrightarrow> Ord(K) & (\<forall>j. j<K \<longrightarrow> ~ j \<approx> K)"
proof -
  { fix j
    assume K: "Card(K)" "j \<approx> K"
    assume "j < K"
    also have "... = (\<mu> i. i \<approx> K)" using K
      by (simp add: Card_def cardinal_def)
    finally have "j < (\<mu> i. i \<approx> K)" .
    hence "False" using K
      by (best dest: less_LeastE)
  }
  then show ?thesis
    by (blast intro: CardI Card_is_Ord)
qed

lemma lt_Card_imp_lesspoll: "[| Card(a); i<a |] ==> i \<prec> a"
apply (unfold lesspoll_def)
apply (drule Card_iff_initial [THEN iffD1])
apply (blast intro!: leI [THEN le_imp_lepoll])
done

lemma Card_0: "Card(0)"
apply (rule Ord_0 [THEN CardI])
apply (blast elim!: ltE)
done

lemma Card_Un: "[| Card(K);  Card(L) |] ==> Card(K \<union> L)"
apply (rule Ord_linear_le [of K L])
apply (simp_all add: subset_Un_iff [THEN iffD1]  Card_is_Ord le_imp_subset
                     subset_Un_iff2 [THEN iffD1])
done

(*Infinite unions of cardinals?  See Devlin, Lemma 6.7, page 98*)

lemma Card_cardinal [iff]: "Card(|A|)"
proof (unfold cardinal_def)
  show "Card(\<mu> i. i \<approx> A)"
    proof (cases "\<exists>i. Ord (i) & i \<approx> A")
      case False thus ?thesis           \<comment>\<open>degenerate case\<close>
        by (simp add: Least_0 Card_0)
    next
      case True                         \<comment>\<open>real case: @{term A} is isomorphic to some ordinal\<close>
      then obtain i where i: "Ord(i)" "i \<approx> A" by blast
      show ?thesis
        proof (rule CardI [OF Ord_Least], rule notI)
          fix j
          assume j: "j < (\<mu> i. i \<approx> A)"
          assume "j \<approx> (\<mu> i. i \<approx> A)"
          also have "... \<approx> A" using i by (auto intro: LeastI)
          finally have "j \<approx> A" .
          thus False
            by (rule less_LeastE [OF _ j])
        qed
    qed
qed

(*Kunen's Lemma 10.5*)
lemma cardinal_eq_lemma:
  assumes i:"|i| \<le> j" and j: "j \<le> i" shows "|j| = |i|"
proof (rule eqpollI [THEN cardinal_cong])
  show "j \<lesssim> i" by (rule le_imp_lepoll [OF j])
next
  have Oi: "Ord(i)" using j by (rule le_Ord2)
  hence "i \<approx> |i|"
    by (blast intro: Ord_cardinal_eqpoll eqpoll_sym)
  also have "... \<lesssim> j"
    by (blast intro: le_imp_lepoll i)
  finally show "i \<lesssim> j" .
qed

lemma cardinal_mono:
  assumes ij: "i \<le> j" shows "|i| \<le> |j|"
using Ord_cardinal [of i] Ord_cardinal [of j]
proof (cases rule: Ord_linear_le)
  case le thus ?thesis .
next
  case ge
  have i: "Ord(i)" using ij
    by (simp add: lt_Ord)
  have ci: "|i| \<le> j"
    by (blast intro: Ord_cardinal_le ij le_trans i)
  have "|i| = ||i||"
    by (auto simp add: Ord_cardinal_idem i)
  also have "... = |j|"
    by (rule cardinal_eq_lemma [OF ge ci])
  finally have "|i| = |j|" .
  thus ?thesis by simp
qed

text\<open>Since we have @{term"|succ(nat)| \<le> |nat|"}, the converse of \<open>cardinal_mono\<close> fails!\<close>
lemma cardinal_lt_imp_lt: "[| |i| < |j|;  Ord(i);  Ord(j) |] ==> i < j"
apply (rule Ord_linear2 [of i j], assumption+)
apply (erule lt_trans2 [THEN lt_irrefl])
apply (erule cardinal_mono)
done

lemma Card_lt_imp_lt: "[| |i| < K;  Ord(i);  Card(K) |] ==> i < K"
  by (simp (no_asm_simp) add: cardinal_lt_imp_lt Card_is_Ord Card_cardinal_eq)

lemma Card_lt_iff: "[| Ord(i);  Card(K) |] ==> (|i| < K) \<longleftrightarrow> (i < K)"
by (blast intro: Card_lt_imp_lt Ord_cardinal_le [THEN lt_trans1])

lemma Card_le_iff: "[| Ord(i);  Card(K) |] ==> (K \<le> |i|) \<longleftrightarrow> (K \<le> i)"
by (simp add: Card_lt_iff Card_is_Ord Ord_cardinal not_lt_iff_le [THEN iff_sym])

(*Can use AC or finiteness to discharge first premise*)
lemma well_ord_lepoll_imp_Card_le:
  assumes wB: "well_ord(B,r)" and AB: "A \<lesssim> B"
  shows "|A| \<le> |B|"
using Ord_cardinal [of A] Ord_cardinal [of B]
proof (cases rule: Ord_linear_le)
  case le thus ?thesis .
next
  case ge
  from lepoll_well_ord [OF AB wB]
  obtain s where s: "well_ord(A, s)" by blast
  have "B  \<approx> |B|" by (blast intro: wB eqpoll_sym well_ord_cardinal_eqpoll)
  also have "... \<lesssim> |A|" by (rule le_imp_lepoll [OF ge])
  also have "... \<approx> A" by (rule well_ord_cardinal_eqpoll [OF s])
  finally have "B \<lesssim> A" .
  hence "A \<approx> B" by (blast intro: eqpollI AB)
  hence "|A| = |B|" by (rule cardinal_cong)
  thus ?thesis by simp
qed

lemma lepoll_cardinal_le: "[| A \<lesssim> i; Ord(i) |] ==> |A| \<le> i"
apply (rule le_trans)
apply (erule well_ord_Memrel [THEN well_ord_lepoll_imp_Card_le], assumption)
apply (erule Ord_cardinal_le)
done

lemma lepoll_Ord_imp_eqpoll: "[| A \<lesssim> i; Ord(i) |] ==> |A| \<approx> A"
by (blast intro: lepoll_cardinal_le well_ord_Memrel well_ord_cardinal_eqpoll dest!: lepoll_well_ord)

lemma lesspoll_imp_eqpoll: "[| A \<prec> i; Ord(i) |] ==> |A| \<approx> A"
apply (unfold lesspoll_def)
apply (blast intro: lepoll_Ord_imp_eqpoll)
done

lemma cardinal_subset_Ord: "[|A<=i; Ord(i)|] ==> |A| \<subseteq> i"
apply (drule subset_imp_lepoll [THEN lepoll_cardinal_le])
apply (auto simp add: lt_def)
apply (blast intro: Ord_trans)
done

subsection\<open>The finite cardinals\<close>

lemma cons_lepoll_consD:
 "[| cons(u,A) \<lesssim> cons(v,B);  u\<notin>A;  v\<notin>B |] ==> A \<lesssim> B"
apply (unfold lepoll_def inj_def, safe)
apply (rule_tac x = "\<lambda>x\<in>A. if f`x=v then f`u else f`x" in exI)
apply (rule CollectI)
(*Proving it's in the function space A->B*)
apply (rule if_type [THEN lam_type])
apply (blast dest: apply_funtype)
apply (blast elim!: mem_irrefl dest: apply_funtype)
(*Proving it's injective*)
apply (simp (no_asm_simp))
apply blast
done

lemma cons_eqpoll_consD: "[| cons(u,A) \<approx> cons(v,B);  u\<notin>A;  v\<notin>B |] ==> A \<approx> B"
apply (simp add: eqpoll_iff)
apply (blast intro: cons_lepoll_consD)
done

(*Lemma suggested by Mike Fourman*)
lemma succ_lepoll_succD: "succ(m) \<lesssim> succ(n) ==> m \<lesssim> n"
apply (unfold succ_def)
apply (erule cons_lepoll_consD)
apply (rule mem_not_refl)+
done

lemma nat_lepoll_imp_le:
     "m \<in> nat ==> n \<in> nat \<Longrightarrow> m \<lesssim> n \<Longrightarrow> m \<le> n"
proof (induct m arbitrary: n rule: nat_induct)
  case 0 thus ?case by (blast intro!: nat_0_le)
next
  case (succ m)
  show ?case  using \<open>n \<in> nat\<close>
    proof (cases rule: natE)
      case 0 thus ?thesis using succ
        by (simp add: lepoll_def inj_def)
    next
      case (succ n') thus ?thesis using succ.hyps \<open> succ(m) \<lesssim> n\<close>
        by (blast intro!: succ_leI dest!: succ_lepoll_succD)
    qed
qed

lemma nat_eqpoll_iff: "[| m \<in> nat; n \<in> nat |] ==> m \<approx> n \<longleftrightarrow> m = n"
apply (rule iffI)
apply (blast intro: nat_lepoll_imp_le le_anti_sym elim!: eqpollE)
apply (simp add: eqpoll_refl)
done

(*The object of all this work: every natural number is a (finite) cardinal*)
lemma nat_into_Card:
  assumes n: "n \<in> nat" shows "Card(n)"
proof (unfold Card_def cardinal_def, rule sym)
  have "Ord(n)" using n  by auto
  moreover
  { fix i
    assume "i < n" "i \<approx> n"
    hence False using n
      by (auto simp add: lt_nat_in_nat [THEN nat_eqpoll_iff])
  }
  ultimately show "(\<mu> i. i \<approx> n) = n" by (auto intro!: Least_equality)
qed

lemmas cardinal_0 = nat_0I [THEN nat_into_Card, THEN Card_cardinal_eq, iff]
lemmas cardinal_1 = nat_1I [THEN nat_into_Card, THEN Card_cardinal_eq, iff]

(*Part of Kunen's Lemma 10.6*)
lemma succ_lepoll_natE: "[| succ(n) \<lesssim> n;  n \<in> nat |] ==> P"
by (rule nat_lepoll_imp_le [THEN lt_irrefl], auto)

lemma nat_lepoll_imp_ex_eqpoll_n:
     "[| n \<in> nat;  nat \<lesssim> X |] ==> \<exists>Y. Y \<subseteq> X & n \<approx> Y"
apply (unfold lepoll_def eqpoll_def)
apply (fast del: subsetI subsetCE
            intro!: subset_SIs
            dest!: Ord_nat [THEN [2] OrdmemD, THEN [2] restrict_inj]
            elim!: restrict_bij
                   inj_is_fun [THEN fun_is_rel, THEN image_subset])
done

(** \<lesssim>, \<prec> and natural numbers **)

lemma lepoll_succ: "i \<lesssim> succ(i)"
  by (blast intro: subset_imp_lepoll)

lemma lepoll_imp_lesspoll_succ:
  assumes A: "A \<lesssim> m" and m: "m \<in> nat"
  shows "A \<prec> succ(m)"
proof -
  { assume "A \<approx> succ(m)"
    hence "succ(m) \<approx> A" by (rule eqpoll_sym)
    also have "... \<lesssim> m" by (rule A)
    finally have "succ(m) \<lesssim> m" .
    hence False by (rule succ_lepoll_natE) (rule m) }
  moreover have "A \<lesssim> succ(m)" by (blast intro: lepoll_trans A lepoll_succ)
  ultimately show ?thesis by (auto simp add: lesspoll_def)
qed

lemma lesspoll_succ_imp_lepoll:
     "[| A \<prec> succ(m); m \<in> nat |] ==> A \<lesssim> m"
apply (unfold lesspoll_def lepoll_def eqpoll_def bij_def)
apply (auto dest: inj_not_surj_succ)
done

lemma lesspoll_succ_iff: "m \<in> nat ==> A \<prec> succ(m) \<longleftrightarrow> A \<lesssim> m"
by (blast intro!: lepoll_imp_lesspoll_succ lesspoll_succ_imp_lepoll)

lemma lepoll_succ_disj: "[| A \<lesssim> succ(m);  m \<in> nat |] ==> A \<lesssim> m | A \<approx> succ(m)"
apply (rule disjCI)
apply (rule lesspoll_succ_imp_lepoll)
prefer 2 apply assumption
apply (simp (no_asm_simp) add: lesspoll_def)
done

lemma lesspoll_cardinal_lt: "[| A \<prec> i; Ord(i) |] ==> |A| < i"
apply (unfold lesspoll_def, clarify)
apply (frule lepoll_cardinal_le, assumption)
apply (blast intro: well_ord_Memrel well_ord_cardinal_eqpoll [THEN eqpoll_sym]
             dest: lepoll_well_ord  elim!: leE)
done

subsection\<open>The first infinite cardinal: Omega, or nat\<close>

(*This implies Kunen's Lemma 10.6*)
lemma lt_not_lepoll:
  assumes n: "n<i" "n \<in> nat" shows "~ i \<lesssim> n"
proof -
  { assume i: "i \<lesssim> n"
    have "succ(n) \<lesssim> i" using n
      by (elim ltE, blast intro: Ord_succ_subsetI [THEN subset_imp_lepoll])
    also have "... \<lesssim> n" by (rule i)
    finally have "succ(n) \<lesssim> n" .
    hence False  by (rule succ_lepoll_natE) (rule n) }
  thus ?thesis by auto
qed

text\<open>A slightly weaker version of \<open>nat_eqpoll_iff\<close>\<close>
lemma Ord_nat_eqpoll_iff:
  assumes i: "Ord(i)" and n: "n \<in> nat" shows "i \<approx> n \<longleftrightarrow> i=n"
using i nat_into_Ord [OF n]
proof (cases rule: Ord_linear_lt)
  case lt
  hence  "i \<in> nat" by (rule lt_nat_in_nat) (rule n)
  thus ?thesis by (simp add: nat_eqpoll_iff n)
next
  case eq
  thus ?thesis by (simp add: eqpoll_refl)
next
  case gt
  hence  "~ i \<lesssim> n" using n  by (rule lt_not_lepoll)
  hence  "~ i \<approx> n" using n  by (blast intro: eqpoll_imp_lepoll)
  moreover have "i \<noteq> n" using \<open>n<i\<close> by auto
  ultimately show ?thesis by blast
qed

lemma Card_nat: "Card(nat)"
proof -
  { fix i
    assume i: "i < nat" "i \<approx> nat"
    hence "~ nat \<lesssim> i"
      by (simp add: lt_def lt_not_lepoll)
    hence False using i
      by (simp add: eqpoll_iff)
  }
  hence "(\<mu> i. i \<approx> nat) = nat" by (blast intro: Least_equality eqpoll_refl)
  thus ?thesis
    by (auto simp add: Card_def cardinal_def)
qed

(*Allows showing that |i| is a limit cardinal*)
lemma nat_le_cardinal: "nat \<le> i ==> nat \<le> |i|"
apply (rule Card_nat [THEN Card_cardinal_eq, THEN subst])
apply (erule cardinal_mono)
done

lemma n_lesspoll_nat: "n \<in> nat ==> n \<prec> nat"
  by (blast intro: Ord_nat Card_nat ltI lt_Card_imp_lesspoll)

subsection\<open>Towards Cardinal Arithmetic\<close>
(** Congruence laws for successor, cardinal addition and multiplication **)

(*Congruence law for  cons  under equipollence*)
lemma cons_lepoll_cong:
    "[| A \<lesssim> B;  b \<notin> B |] ==> cons(a,A) \<lesssim> cons(b,B)"
apply (unfold lepoll_def, safe)
apply (rule_tac x = "\<lambda>y\<in>cons (a,A) . if y=a then b else f`y" in exI)
apply (rule_tac d = "%z. if z \<in> B then converse (f) `z else a" in lam_injective)
apply (safe elim!: consE')
   apply simp_all
apply (blast intro: inj_is_fun [THEN apply_type])+
done

lemma cons_eqpoll_cong:
     "[| A \<approx> B;  a \<notin> A;  b \<notin> B |] ==> cons(a,A) \<approx> cons(b,B)"
by (simp add: eqpoll_iff cons_lepoll_cong)

lemma cons_lepoll_cons_iff:
     "[| a \<notin> A;  b \<notin> B |] ==> cons(a,A) \<lesssim> cons(b,B)  \<longleftrightarrow>  A \<lesssim> B"
by (blast intro: cons_lepoll_cong cons_lepoll_consD)

lemma cons_eqpoll_cons_iff:
     "[| a \<notin> A;  b \<notin> B |] ==> cons(a,A) \<approx> cons(b,B)  \<longleftrightarrow>  A \<approx> B"
by (blast intro: cons_eqpoll_cong cons_eqpoll_consD)

lemma singleton_eqpoll_1: "{a} \<approx> 1"
apply (unfold succ_def)
apply (blast intro!: eqpoll_refl [THEN cons_eqpoll_cong])
done

lemma cardinal_singleton: "|{a}| = 1"
apply (rule singleton_eqpoll_1 [THEN cardinal_cong, THEN trans])
apply (simp (no_asm) add: nat_into_Card [THEN Card_cardinal_eq])
done

lemma not_0_is_lepoll_1: "A \<noteq> 0 ==> 1 \<lesssim> A"
apply (erule not_emptyE)
apply (rule_tac a = "cons (x, A-{x}) " in subst)
apply (rule_tac [2] a = "cons(0,0)" and P= "%y. y \<lesssim> cons (x, A-{x})" in subst)
prefer 3 apply (blast intro: cons_lepoll_cong subset_imp_lepoll, auto)
done

(*Congruence law for  succ  under equipollence*)
lemma succ_eqpoll_cong: "A \<approx> B ==> succ(A) \<approx> succ(B)"
apply (unfold succ_def)
apply (simp add: cons_eqpoll_cong mem_not_refl)
done

(*Congruence law for + under equipollence*)
lemma sum_eqpoll_cong: "[| A \<approx> C;  B \<approx> D |] ==> A+B \<approx> C+D"
apply (unfold eqpoll_def)
apply (blast intro!: sum_bij)
done

(*Congruence law for * under equipollence*)
lemma prod_eqpoll_cong:
    "[| A \<approx> C;  B \<approx> D |] ==> A*B \<approx> C*D"
apply (unfold eqpoll_def)
apply (blast intro!: prod_bij)
done

lemma inj_disjoint_eqpoll:
    "[| f \<in> inj(A,B);  A \<inter> B = 0 |] ==> A \<union> (B - range(f)) \<approx> B"
apply (unfold eqpoll_def)
apply (rule exI)
apply (rule_tac c = "%x. if x \<in> A then f`x else x"
            and d = "%y. if y \<in> range (f) then converse (f) `y else y"
       in lam_bijective)
apply (blast intro!: if_type inj_is_fun [THEN apply_type])
apply (simp (no_asm_simp) add: inj_converse_fun [THEN apply_funtype])
apply (safe elim!: UnE')
   apply (simp_all add: inj_is_fun [THEN apply_rangeI])
apply (blast intro: inj_converse_fun [THEN apply_type])+
done

subsection\<open>Lemmas by Krzysztof Grabczewski\<close>

(*New proofs using cons_lepoll_cons. Could generalise from succ to cons.*)

text\<open>If @{term A} has at most @{term"n+1"} elements and @{term"a \<in> A"}
      then @{term"A-{a}"} has at most @{term n}.\<close>
lemma Diff_sing_lepoll:
      "[| a \<in> A;  A \<lesssim> succ(n) |] ==> A - {a} \<lesssim> n"
apply (unfold succ_def)
apply (rule cons_lepoll_consD)
apply (rule_tac [3] mem_not_refl)
apply (erule cons_Diff [THEN ssubst], safe)
done

text\<open>If @{term A} has at least @{term"n+1"} elements then @{term"A-{a}"} has at least @{term n}.\<close>
lemma lepoll_Diff_sing:
  assumes A: "succ(n) \<lesssim> A" shows "n \<lesssim> A - {a}"
proof -
  have "cons(n,n) \<lesssim> A" using A
    by (unfold succ_def)
  also have "... \<lesssim> cons(a, A-{a})"
    by (blast intro: subset_imp_lepoll)
  finally have "cons(n,n) \<lesssim> cons(a, A-{a})" .
  thus ?thesis
    by (blast intro: cons_lepoll_consD mem_irrefl)
qed

lemma Diff_sing_eqpoll: "[| a \<in> A; A \<approx> succ(n) |] ==> A - {a} \<approx> n"
by (blast intro!: eqpollI
          elim!: eqpollE
          intro: Diff_sing_lepoll lepoll_Diff_sing)

lemma lepoll_1_is_sing: "[| A \<lesssim> 1; a \<in> A |] ==> A = {a}"
apply (frule Diff_sing_lepoll, assumption)
apply (drule lepoll_0_is_0)
apply (blast elim: equalityE)
done

lemma Un_lepoll_sum: "A \<union> B \<lesssim> A+B"
apply (unfold lepoll_def)
apply (rule_tac x = "\<lambda>x\<in>A \<union> B. if x\<in>A then Inl (x) else Inr (x)" in exI)
apply (rule_tac d = "%z. snd (z)" in lam_injective)
apply force
apply (simp add: Inl_def Inr_def)
done

lemma well_ord_Un:
     "[| well_ord(X,R); well_ord(Y,S) |] ==> \<exists>T. well_ord(X \<union> Y, T)"
by (erule well_ord_radd [THEN Un_lepoll_sum [THEN lepoll_well_ord]],
    assumption)

(*Krzysztof Grabczewski*)
lemma disj_Un_eqpoll_sum: "A \<inter> B = 0 ==> A \<union> B \<approx> A + B"
apply (unfold eqpoll_def)
apply (rule_tac x = "\<lambda>a\<in>A \<union> B. if a \<in> A then Inl (a) else Inr (a)" in exI)
apply (rule_tac d = "%z. case (%x. x, %x. x, z)" in lam_bijective)
apply auto
done

subsection \<open>Finite and infinite sets\<close>

lemma eqpoll_imp_Finite_iff: "A \<approx> B ==> Finite(A) \<longleftrightarrow> Finite(B)"
apply (unfold Finite_def)
apply (blast intro: eqpoll_trans eqpoll_sym)
done

lemma Finite_0 [simp]: "Finite(0)"
apply (unfold Finite_def)
apply (blast intro!: eqpoll_refl nat_0I)
done

lemma Finite_cons: "Finite(x) ==> Finite(cons(y,x))"
apply (unfold Finite_def)
apply (case_tac "y \<in> x")
apply (simp add: cons_absorb)
apply (erule bexE)
apply (rule bexI)
apply (erule_tac [2] nat_succI)
apply (simp (no_asm_simp) add: succ_def cons_eqpoll_cong mem_not_refl)
done

lemma Finite_succ: "Finite(x) ==> Finite(succ(x))"
apply (unfold succ_def)
apply (erule Finite_cons)
done

lemma lepoll_nat_imp_Finite:
  assumes A: "A \<lesssim> n" and n: "n \<in> nat" shows "Finite(A)"
proof -
  have "A \<lesssim> n \<Longrightarrow> Finite(A)" using n
    proof (induct n)
      case 0
      hence "A = 0" by (rule lepoll_0_is_0)
      thus ?case by simp
    next
      case (succ n)
      hence "A \<lesssim> n \<or> A \<approx> succ(n)" by (blast dest: lepoll_succ_disj)
      thus ?case using succ by (auto simp add: Finite_def)
    qed
  thus ?thesis using A .
qed

lemma lesspoll_nat_is_Finite:
     "A \<prec> nat ==> Finite(A)"
apply (unfold Finite_def)
apply (blast dest: ltD lesspoll_cardinal_lt
                   lesspoll_imp_eqpoll [THEN eqpoll_sym])
done

lemma lepoll_Finite:
  assumes Y: "Y \<lesssim> X" and X: "Finite(X)" shows "Finite(Y)"
proof -
  obtain n where n: "n \<in> nat" "X \<approx> n" using X
    by (auto simp add: Finite_def)
  have "Y \<lesssim> X"         by (rule Y)
  also have "... \<approx> n"  by (rule n)
  finally have "Y \<lesssim> n" .
  thus ?thesis using n by (simp add: lepoll_nat_imp_Finite)
qed

lemmas subset_Finite = subset_imp_lepoll [THEN lepoll_Finite]

lemma Finite_cons_iff [iff]: "Finite(cons(y,x)) \<longleftrightarrow> Finite(x)"
by (blast intro: Finite_cons subset_Finite)

lemma Finite_succ_iff [iff]: "Finite(succ(x)) \<longleftrightarrow> Finite(x)"
by (simp add: succ_def)

lemma Finite_Int: "Finite(A) | Finite(B) ==> Finite(A \<inter> B)"
by (blast intro: subset_Finite)

lemmas Finite_Diff = Diff_subset [THEN subset_Finite]

lemma nat_le_infinite_Ord:
      "[| Ord(i);  ~ Finite(i) |] ==> nat \<le> i"
apply (unfold Finite_def)
apply (erule Ord_nat [THEN [2] Ord_linear2])
prefer 2 apply assumption
apply (blast intro!: eqpoll_refl elim!: ltE)
done

lemma Finite_imp_well_ord:
    "Finite(A) ==> \<exists>r. well_ord(A,r)"
apply (unfold Finite_def eqpoll_def)
apply (blast intro: well_ord_rvimage bij_is_inj well_ord_Memrel nat_into_Ord)
done

lemma succ_lepoll_imp_not_empty: "succ(x) \<lesssim> y ==> y \<noteq> 0"
by (fast dest!: lepoll_0_is_0)

lemma eqpoll_succ_imp_not_empty: "x \<approx> succ(n) ==> x \<noteq> 0"
by (fast elim!: eqpoll_sym [THEN eqpoll_0_is_0, THEN succ_neq_0])

lemma Finite_Fin_lemma [rule_format]:
     "n \<in> nat ==> \<forall>A. (A\<approx>n & A \<subseteq> X) \<longrightarrow> A \<in> Fin(X)"
apply (induct_tac n)
apply (rule allI)
apply (fast intro!: Fin.emptyI dest!: eqpoll_imp_lepoll [THEN lepoll_0_is_0])
apply (rule allI)
apply (rule impI)
apply (erule conjE)
apply (rule eqpoll_succ_imp_not_empty [THEN not_emptyE], assumption)
apply (frule Diff_sing_eqpoll, assumption)
apply (erule allE)
apply (erule impE, fast)
apply (drule subsetD, assumption)
apply (drule Fin.consI, assumption)
apply (simp add: cons_Diff)
done

lemma Finite_Fin: "[| Finite(A); A \<subseteq> X |] ==> A \<in> Fin(X)"
by (unfold Finite_def, blast intro: Finite_Fin_lemma)

lemma Fin_lemma [rule_format]: "n \<in> nat ==> \<forall>A. A \<approx> n \<longrightarrow> A \<in> Fin(A)"
apply (induct_tac n)
apply (simp add: eqpoll_0_iff, clarify)
apply (subgoal_tac "\<exists>u. u \<in> A")
apply (erule exE)
apply (rule Diff_sing_eqpoll [elim_format])
prefer 2 apply assumption
apply assumption
apply (rule_tac b = A in cons_Diff [THEN subst], assumption)
apply (rule Fin.consI, blast)
apply (blast intro: subset_consI [THEN Fin_mono, THEN subsetD])
(*Now for the lemma assumed above*)
apply (unfold eqpoll_def)
apply (blast intro: bij_converse_bij [THEN bij_is_fun, THEN apply_type])
done

lemma Finite_into_Fin: "Finite(A) ==> A \<in> Fin(A)"
apply (unfold Finite_def)
apply (blast intro: Fin_lemma)
done

lemma Fin_into_Finite: "A \<in> Fin(U) ==> Finite(A)"
by (fast intro!: Finite_0 Finite_cons elim: Fin_induct)

lemma Finite_Fin_iff: "Finite(A) \<longleftrightarrow> A \<in> Fin(A)"
by (blast intro: Finite_into_Fin Fin_into_Finite)

lemma Finite_Un: "[| Finite(A); Finite(B) |] ==> Finite(A \<union> B)"
by (blast intro!: Fin_into_Finite Fin_UnI
          dest!: Finite_into_Fin
          intro: Un_upper1 [THEN Fin_mono, THEN subsetD]
                 Un_upper2 [THEN Fin_mono, THEN subsetD])

lemma Finite_Un_iff [simp]: "Finite(A \<union> B) \<longleftrightarrow> (Finite(A) & Finite(B))"
by (blast intro: subset_Finite Finite_Un)

text\<open>The converse must hold too.\<close>
lemma Finite_Union: "[| \<forall>y\<in>X. Finite(y);  Finite(X) |] ==> Finite(\<Union>(X))"
apply (simp add: Finite_Fin_iff)
apply (rule Fin_UnionI)
apply (erule Fin_induct, simp)
apply (blast intro: Fin.consI Fin_mono [THEN [2] rev_subsetD])
done

(* Induction principle for Finite(A), by Sidi Ehmety *)
lemma Finite_induct [case_names 0 cons, induct set: Finite]:
"[| Finite(A); P(0);
    !! x B.   [| Finite(B); x \<notin> B; P(B) |] ==> P(cons(x, B)) |]
 ==> P(A)"
apply (erule Finite_into_Fin [THEN Fin_induct])
apply (blast intro: Fin_into_Finite)+
done

(*Sidi Ehmety.  The contrapositive says ~Finite(A) ==> ~Finite(A-{a}) *)
lemma Diff_sing_Finite: "Finite(A - {a}) ==> Finite(A)"
apply (unfold Finite_def)
apply (case_tac "a \<in> A")
apply (subgoal_tac [2] "A-{a}=A", auto)
apply (rule_tac x = "succ (n) " in bexI)
apply (subgoal_tac "cons (a, A - {a}) = A & cons (n, n) = succ (n) ")
apply (drule_tac a = a and b = n in cons_eqpoll_cong)
apply (auto dest: mem_irrefl)
done

(*Sidi Ehmety.  And the contrapositive of this says
   [| ~Finite(A); Finite(B) |] ==> ~Finite(A-B) *)
lemma Diff_Finite [rule_format]: "Finite(B) ==> Finite(A-B) \<longrightarrow> Finite(A)"
apply (erule Finite_induct, auto)
apply (case_tac "x \<in> A")
 apply (subgoal_tac [2] "A-cons (x, B) = A - B")
apply (subgoal_tac "A - cons (x, B) = (A - B) - {x}", simp)
apply (drule Diff_sing_Finite, auto)
done

lemma Finite_RepFun: "Finite(A) ==> Finite(RepFun(A,f))"
by (erule Finite_induct, simp_all)

lemma Finite_RepFun_iff_lemma [rule_format]:
     "[|Finite(x); !!x y. f(x)=f(y) ==> x=y|]
      ==> \<forall>A. x = RepFun(A,f) \<longrightarrow> Finite(A)"
apply (erule Finite_induct)
 apply clarify
 apply (case_tac "A=0", simp)
 apply (blast del: allE, clarify)
apply (subgoal_tac "\<exists>z\<in>A. x = f(z)")
 prefer 2 apply (blast del: allE elim: equalityE, clarify)
apply (subgoal_tac "B = {f(u) . u \<in> A - {z}}")
 apply (blast intro: Diff_sing_Finite)
apply (thin_tac "\<forall>A. P(A) \<longrightarrow> Finite(A)" for P)
apply (rule equalityI)
 apply (blast intro: elim: equalityE)
apply (blast intro: elim: equalityCE)
done

text\<open>I don't know why, but if the premise is expressed using meta-connectives
then  the simplifier cannot prove it automatically in conditional rewriting.\<close>
lemma Finite_RepFun_iff:
     "(\<forall>x y. f(x)=f(y) \<longrightarrow> x=y) ==> Finite(RepFun(A,f)) \<longleftrightarrow> Finite(A)"
by (blast intro: Finite_RepFun Finite_RepFun_iff_lemma [of _ f])

lemma Finite_Pow: "Finite(A) ==> Finite(Pow(A))"
apply (erule Finite_induct)
apply (simp_all add: Pow_insert Finite_Un Finite_RepFun)
done

lemma Finite_Pow_imp_Finite: "Finite(Pow(A)) ==> Finite(A)"
apply (subgoal_tac "Finite({{x} . x \<in> A})")
 apply (simp add: Finite_RepFun_iff )
apply (blast intro: subset_Finite)
done

lemma Finite_Pow_iff [iff]: "Finite(Pow(A)) \<longleftrightarrow> Finite(A)"
by (blast intro: Finite_Pow Finite_Pow_imp_Finite)

lemma Finite_cardinal_iff:
  assumes i: "Ord(i)" shows "Finite(|i|) \<longleftrightarrow> Finite(i)"
  by (auto simp add: Finite_def) (blast intro: eqpoll_trans eqpoll_sym Ord_cardinal_eqpoll [OF i])+

(*Krzysztof Grabczewski's proof that the converse of a finite, well-ordered
  set is well-ordered.  Proofs simplified by lcp. *)

lemma nat_wf_on_converse_Memrel: "n \<in> nat ==> wf[n](converse(Memrel(n)))"
proof (induct n rule: nat_induct)
  case 0 thus ?case by (blast intro: wf_onI)
next
  case (succ x)
  hence wfx: "\<And>Z. Z = 0 \<or> (\<exists>z\<in>Z. \<forall>y. z \<in> y \<and> z \<in> x \<and> y \<in> x \<and> z \<in> x \<longrightarrow> y \<notin> Z)"
    by (simp add: wf_on_def wf_def)  \<comment>\<open>not easy to erase the duplicate @{term"z \<in> x"}!\<close>
  show ?case
    proof (rule wf_onI)
      fix Z u
      assume Z: "u \<in> Z" "\<forall>z\<in>Z. \<exists>y\<in>Z. \<langle>y, z\<rangle> \<in> converse(Memrel(succ(x)))"
      show False
        proof (cases "x \<in> Z")
          case True thus False using Z
            by (blast elim: mem_irrefl mem_asym)
          next
          case False thus False using wfx [of Z] Z
            by blast
        qed
    qed
qed

lemma nat_well_ord_converse_Memrel: "n \<in> nat ==> well_ord(n,converse(Memrel(n)))"
apply (frule Ord_nat [THEN Ord_in_Ord, THEN well_ord_Memrel])
apply (simp add: well_ord_def tot_ord_converse nat_wf_on_converse_Memrel)
done

lemma well_ord_converse:
     "[|well_ord(A,r);
        well_ord(ordertype(A,r), converse(Memrel(ordertype(A, r)))) |]
      ==> well_ord(A,converse(r))"
apply (rule well_ord_Int_iff [THEN iffD1])
apply (frule ordermap_bij [THEN bij_is_inj, THEN well_ord_rvimage], assumption)
apply (simp add: rvimage_converse converse_Int converse_prod
                 ordertype_ord_iso [THEN ord_iso_rvimage_eq])
done

lemma ordertype_eq_n:
  assumes r: "well_ord(A,r)" and A: "A \<approx> n" and n: "n \<in> nat"
  shows "ordertype(A,r) = n"
proof -
  have "ordertype(A,r) \<approx> A"
    by (blast intro: bij_imp_eqpoll bij_converse_bij ordermap_bij r)
  also have "... \<approx> n" by (rule A)
  finally have "ordertype(A,r) \<approx> n" .
  thus ?thesis
    by (simp add: Ord_nat_eqpoll_iff Ord_ordertype n r)
qed

lemma Finite_well_ord_converse:
    "[| Finite(A);  well_ord(A,r) |] ==> well_ord(A,converse(r))"
apply (unfold Finite_def)
apply (rule well_ord_converse, assumption)
apply (blast dest: ordertype_eq_n intro!: nat_well_ord_converse_Memrel)
done

lemma nat_into_Finite: "n \<in> nat ==> Finite(n)"
  by (auto simp add: Finite_def intro: eqpoll_refl)

lemma nat_not_Finite: "~ Finite(nat)"
proof -
  { fix n
    assume n: "n \<in> nat" "nat \<approx> n"
    have "n \<in> nat"    by (rule n)
    also have "... = n" using n
      by (simp add: Ord_nat_eqpoll_iff Ord_nat)
    finally have "n \<in> n" .
    hence False
      by (blast elim: mem_irrefl)
  }
  thus ?thesis
    by (auto simp add: Finite_def)
qed

end


theory List_extra
imports
  Main
  "HOL.List"
  "List-Index.List_Index"
  "HOL-Eisbach.Eisbach"
  "HOL-Eisbach.Eisbach_Tools"
begin

text{*
  This theory contains all references to list iterators and extra functions (like replace,...) not present in the commun list library and
  the AFP list-index library
*}

fun list_iter ::"('a\<Rightarrow>'b \<Rightarrow> 'b) \<Rightarrow> 'b \<Rightarrow> 'a list \<Rightarrow> 'b" where
"list_iter f r [] = r" |
"list_iter f r (a#xs) = f a (list_iter f r xs)"

fun replace ::"nat \<Rightarrow> 'a \<Rightarrow> 'a list \<Rightarrow> 'a list" where
"replace n x xs =
  (if length xs \<le> n then xs
    else (take n xs) @ [x] @ (drop (Suc n) xs))"

abbreviation fst_extract::"('a\<times>'b) list \<Rightarrow> 'a list" where
"fst_extract L \<equiv> (list_iter (\<lambda>p r. fst p # r) [] L)"

abbreviation snd_extract::"('a\<times>'b) list \<Rightarrow> 'b list" where
"snd_extract L \<equiv> (list_iter (\<lambda>p r. snd p # r) [] L)"

abbreviation update_snd::"('b \<Rightarrow> 'c) \<Rightarrow> ('a\<times>'b) list \<Rightarrow> ('a\<times>'c) list" where
"update_snd f L \<equiv> zip (fst_extract L) (map f (snd_extract L))"

fun BigCirc::"('a\<Rightarrow>'a) list \<Rightarrow> ('a\<Rightarrow>'a)" ("\<Odot> (_)" [75] 55) where
"\<Odot> [] = id" |
"\<Odot> (f#fs) = f \<circ> (\<Odot> fs)"

fun BigCircT::"('a\<rightharpoonup>'b) list \<Rightarrow> ('a\<rightharpoonup>'b)" ("\<Odot>\<^sub>T (_)" [75] 55) where
"\<Odot>\<^sub>T [] = Map.empty" |
"\<Odot>\<^sub>T (f#fs) = f ++ (\<Odot>\<^sub>T fs)"

fun indexed_map::"nat \<Rightarrow> (nat\<Rightarrow>'a\<Rightarrow>'b) \<Rightarrow> 'a list \<Rightarrow> 'b list" where
"indexed_map n f [] = []" |
"indexed_map n f (x#xs) = f n x # (indexed_map (Suc n) f xs)"

lemma indexed_map_cong[fundef_cong]:
  "n=n1 \<Longrightarrow> L = L1 \<Longrightarrow>(\<And>x m. x \<in> set L1 \<Longrightarrow> f m x = g m x) \<Longrightarrow>
    indexed_map n f L = indexed_map n1 g L1"
by (induction L arbitrary: n n1 L1, auto)

lemma indexed_to_map:
 "indexed_map i f L =
  map (\<lambda>p. f (fst p) (snd p)) (zip [i..<(i+length L)] L)"
proof (induction L arbitrary: i)
  case (Cons a L')
    have "map (\<lambda>p. f (fst p) (snd p)) (zip [i..<(i + length (a # L'))] (a # L')) =
          f i a # map (\<lambda>p. f (fst p) (snd p)) (zip [Suc i..<(Suc i + length L')] L')"
      by (metis (no_types, lifting) enumerate_eq_zip enumerate_simps(2) fst_conv
                map_eq_Cons_conv snd_conv)
    thus ?case
      using Cons[of "Suc i"]
      by force
qed auto

lemma indexed_map_0_len[simp]:
  "length (indexed_map 0 f L) = length L"
by (force simp: indexed_to_map[of 0 f L])

lemma nth_indexed_map_0[simp]:
  "i<length L \<Longrightarrow> indexed_map 0 f L ! i = f i (L!i)"
by (simp add:indexed_to_map[of 0 f L])

lemma replace_inv_length[simp]:
  "length (replace n x S) = length S"
by(induction S arbitrary: x n, auto)

lemma nth_replace[simp]:
  "i<length L \<Longrightarrow> (replace n x L)!i = (if i=n \<and> n<length L then x else (L!i))"
unfolding "replace.simps"
proof (induction L arbitrary: i n)
  case (Cons l L')
    have "n=0 \<Longrightarrow> ?case" by simp
    moreover have "\<And>n1. n=Suc n1 \<Longrightarrow> length (l # L') \<le> n \<Longrightarrow> ?case"
      by simp
    moreover have "\<And>n1. n=Suc n1 \<Longrightarrow>\<not> length (l # L') \<le> n \<Longrightarrow> i = Suc n1 \<Longrightarrow> ?case"
      using Cons(1)[of "i-1" "n-1", simplified]
      by fastforce
    moreover have "\<And>n1. n=Suc n1 \<Longrightarrow>\<not> length (l # L') \<le> n \<Longrightarrow> i \<noteq> Suc n1 \<Longrightarrow> ?case"
      using Cons(1)[of "i-1" "n-1", simplified]
            Cons(2)
      by (cases i, simp+)

    ultimately show ?case by (metis old.nat.exhaust)
qed simp

lemma insert_nth_comp:
  "n\<le> length L \<Longrightarrow> n\<le>n1 \<Longrightarrow> insert_nth n S (insert_nth n1 S1 L) = insert_nth (Suc n1) S1 (insert_nth n S L)"
  "n\<le> length L \<Longrightarrow> n\<ge>n1 \<Longrightarrow> insert_nth (Suc n) S (insert_nth n1 S1 L) = insert_nth (n1) S1 (insert_nth n S L)"
proof (induction L arbitrary: n n1)
  case (Cons a L')
    case (1)
      have "Suc n1 - n = Suc (n1-n)" using 1(2) by auto
      hence "S # drop n (take n1 (a # L')) @ S1 # drop n1 (a # L') = take (Suc n1 - n) (S # drop n (a # L'))
                @ S1 # drop (Suc n1 - n) (S # drop n (a # L'))"
        using take_Cons[of "Suc n1 - n" S] drop_Cons[of "Suc n1 - n" S] drop_drop[of "n1-n" n]
               drop_take[of n n1 "a#L'"]
        by fastforce
      then show ?case using 1 by (simp add:min_def)

next
  case (Cons a L')
    case (2)
      have "Suc n - n1 = Suc (n-n1)" using 2(2) by auto
      hence "take (Suc n - n1) (S1 # drop n1 (a # L')) @ S # drop (Suc n - n1) (S1 # drop n1 (a # L')) =
            S1 # drop n1 (take n (a # L')) @ S # drop n (a # L')"
        using  take_Cons[of "Suc n - n1" S1] drop_Cons[of "Suc n - n1" S1] drop_drop[of "n-n1" n1]
                   drop_take[of n1 n "a#L'"]
        by fastforce
      then show ?case using 2 by (simp add: min_def)
qed simp+

lemma rep_ins:
  "n\<le>n1 \<Longrightarrow> n\<le> length W \<Longrightarrow> insert_nth n S (replace n1 A W) = replace (Suc n1) A (insert_nth n S W)" (is "?P\<Longrightarrow> ?R \<Longrightarrow> ?Q")
proof -
  assume H: "?R" "?P"
  have 1:"n1\<ge> length W \<Longrightarrow> ?Q"
    by (simp add: min_def)
  have "n1< length W \<Longrightarrow> ?Q"
    proof -
      assume H1: "n1<length W"
      have "(Suc (Suc n1) - n) = Suc (Suc n1 - n)"
            using H
            by fastforce
      with H show ?thesis
        by (simp add: H1 min_def)(simp add: H Suc_diff_le take_drop)
    qed
  with 1 show "?Q" by linarith
qed

lemma rep_ins2:
  "n<n1 \<Longrightarrow> n1\<le> length W \<Longrightarrow> insert_nth n1 S (replace n A W) = replace n A (insert_nth n1 S W)" (is "?P\<Longrightarrow> ?R \<Longrightarrow> ?Q")
proof -
  assume H: "?R" "?P"
  have "Suc n \<le> n1"
        by (metis (no_types) H(2) Suc_leI)
  with H show "?Q"
    by (simp add: drop_Cons' drop_take take_Cons' min_def)
qed

lemma len_fst_extract[simp]:
  "length (fst_extract L) = length L"
by (induction L, auto)

lemma len_snd_extract[simp]:
  "length (snd_extract L) = length L"
by (induction L, auto)

lemma fst_extract_zip1:
  "length L = length L1 \<Longrightarrow> fst_extract (zip L L1) = L"
proof(induction L arbitrary: L1)
  case (Cons a L')
    obtain b L1' where "L1 = b#L1'"
      using Cons(2)
      by (metis length_Suc_conv)

    with Cons
      show ?case
        using fst_conv length_Cons list_iter.simps(2) nat.simps(1) zip_Cons_Cons
          by auto
qed auto

lemma zip_comp:
  "L = zip (fst_extract L) (snd_extract L)"
by(induction L,auto)

lemma inset_find_Some:
  "x \<in> set (fst_extract L) \<Longrightarrow> \<exists>p. find (\<lambda>e. fst e = x) L = Some (x,p) \<and> (x,p) \<in> set L"
by (induction L arbitrary: x, auto)

lemma incl_fst:
  "set L \<subseteq> set L1 \<Longrightarrow> set (fst_extract L) \<subseteq> set (fst_extract L1)"
using zip_comp[of L1]
      set_zip_leftD[of _ _ "fst_extract L1" "snd_extract L1"]
by (induction L arbitrary: L1, auto)

lemma incl_snd:
  "set L \<subseteq> set L1 \<Longrightarrow> set (snd_extract L) \<subseteq> set (snd_extract L1)"
using zip_comp[of L1]
      set_zip_rightD[of _ _ "fst_extract L1" "snd_extract L1"]
by (induction L arbitrary: L1, auto)

lemma find_zip1:
  "distinct L \<Longrightarrow> length L = length L1 \<Longrightarrow> length (L@L3)  = length L2 \<Longrightarrow> j< length L \<Longrightarrow> find (\<lambda>p. fst p = k) (zip L L1) = Some ((zip L L1) ! j)
   \<Longrightarrow> find (\<lambda>p. fst p = k) (zip (L@L3) L2) =  Some ((zip (L@L3) L2) ! j)"
proof (induction L arbitrary: k j L1 L2 L3)
  case (Cons a L')
   from this(3,6)
     have "j=0 \<Longrightarrow> a = k"
       using find_Some_iff[of "\<lambda>p. fst p = k" "zip (a#L') L1" "(a,L1!j)"]
             nth_zip[of 0 "a#L'" L1]
             fst_conv length_greater_0_conv list.simps(3) nth_Cons_0
       by fastforce
   with Cons(3,4)
     have H:"j=0 \<Longrightarrow> find (\<lambda>p. fst p = k) (zip (a#L'@L3) L2) =  Some (a,L2 ! j)"
       using find_Some_iff[of "\<lambda>p. fst p = k" "zip (a#L'@L3) L2" "(a,L2!j)"]
             nth_zip[of 0 "a#L'@L3" L2]
             fst_conv length_greater_0_conv list.simps(3) nth_Cons_0
     by fastforce
   obtain b L2' where 1:"L2 = b#L2'"
     using Cons(3,4) length_Suc_conv[of L2 "length L' + length L3"]
           length_append[of "a#L'" L3]
     by auto
   obtain b1 L1' where 2:"L1 = b1#L1'"
     using Cons(2,3) length_Suc_conv
     by metis
   have H2:"j>0 \<Longrightarrow> find (\<lambda>p. fst p = k) (zip (a#L'@L3) L2) =  Some (zip (a # L'@L3) L2 ! j)"
     proof -
       let ?f = "\<lambda>p. fst p = k"
       assume C:"j>0"
       from Cons(2)
         have D:"distinct (zip L' L1') \<and> distinct (zip (a#L') (b1#L1'))"
           by (metis distinct.simps(2) distinct_zipI1)
       from Cons(3,5)
         have C': "j<length (zip (a#L') L1)"
           by auto
       from C obtain p where C1:"j = Suc p"
         using Suc_pred' by blast
       {
          have A:"find ?f (zip (a#L') (b1#L1')) = Some (zip L' L1' ! p)"
            using C1 Cons(6) 2
                  nth_zip[of "Suc p"]
            by fastforce
          hence "Suc p<length (zip (a # L') (b1 # L1')) \<and> fst (zip (a # L') (b1 # L1') ! Suc p) = k \<and>
                zip L' L1' ! p = zip (a # L') (b1 # L1') ! Suc p \<and> (\<forall>j<Suc p. fst (zip (a # L') (b1 # L1') ! j) \<noteq> k)"
            using find_Some_iff[of "?f" "zip (a#L') (b1#L1')" "zip L' L1' ! p"]
                  D C C' C1 2
            by (metis Cons.prems(5) nth_eq_iff_index_eq option.inject)

          hence "p<length (zip L' L1') \<and> fst (zip L' L1' ! p) = k \<and> zip L' L1' ! p = zip L' L1' ! p \<and> (\<forall>j<p. fst (zip L' L1' ! j) \<noteq> k)"
            by auto
          with A have "find ?f (zip (a#L') (b1#L1')) = find ?f (zip L' L1') \<and> find ?f (zip L' L1') = Some (zip L' L1' ! p) "
            using find_Some_iff[of "?f" "zip L' L1'" "zip L' L1' ! p"]
            by metis
       }
       thus ?thesis
         using C C' C1 Cons(2,3,4) 1 2 nth_zip[of "Suc p"]
               Cons(1)[of L1' L3  L2' p]
         by auto
     qed
   show ?case
     proof (cases "j=0")
       case True
         with H show ?thesis
           using nth_Cons_0[of a "L'@L3"] Cons(3,4)
                 length_greater_0_conv[of L2] length_append
                 list.simps(3)[of a "L'@L3"] nth_zip[of 0 "a#L'@L3" L2]
           by fastforce
     next
       case False
         with H2 show ?thesis
           by fastforce
     qed
qed auto

lemma list_iter_nil:
  "list_iter (@) [] L = [] \<Longrightarrow> i<length L \<Longrightarrow> L!i = []"
proof (induction L arbitrary: i)
  case (Cons a L')
    show ?case
      proof (cases "i>0")
        case True
          from this obtain j where "i = Suc j"
            by (metis Suc_pred)
          thus ?thesis
            using Cons(1)[of j]
                  Cons(2,3)
            by force
      next
        case False
          thus ?thesis
            using Cons(2)
            by auto
      qed
qed auto

lemma list_map_incl:
  "set (list_iter (@) [] (map f L)) \<subseteq> S \<Longrightarrow>  i<length L \<Longrightarrow> set(f (L!i)) \<subseteq> S"
proof (induction L arbitrary: f S i)
  case (Cons a L')
    from Cons show ?case
      by (induction i arbitrary: L', auto)
qed auto

lemma list_map_incl2:
  "(\<And>i. i<length L \<Longrightarrow> set(f (L!i)) \<subseteq> S) \<Longrightarrow> set (list_iter (@) [] (map f L)) \<subseteq> S"
proof (induction "map f L" arbitrary: L f S)
  case (Cons a L')
    obtain b L1 where "L = b#L1"
      using Cons(2)
      by blast
    thus ?case
      using Cons(1)[of f L1]
            Cons(3)[of "Suc _"]
            Cons(3)[of 0]
            Cons(2)
      by auto
qed auto

lemma fst_extract_app[simp]:
  "fst_extract (L@L1) = fst_extract L @ fst_extract L1"
by (induction L arbitrary: L1, auto)

lemma snd_extract_app[simp]:
  "snd_extract (L@L1) = snd_extract L @ snd_extract L1"
by (induction L arbitrary: L1, auto)

lemma fst_extract_fst_index[simp]:
  "i< length L \<Longrightarrow> fst_extract L ! i = fst (L ! i)"
proof (induction L arbitrary: i)
  case (Cons a L')
    thus ?case
      by (induction i arbitrary: L', auto)
qed auto

lemma snd_extract_snd_index[simp]:
  "i< length L \<Longrightarrow> snd_extract L ! i = snd (L ! i)"
proof (induction L arbitrary: i)
  case (Cons a L')
    thus ?case
      by (induction i arbitrary: L', auto)
qed auto

lemma fst_updt_snd_is_fst[simp]:
  "fst_extract (update_snd f L) = fst_extract L"
by (induction L arbitrary:f, auto)

lemma fst_ext_com_list_it_app:
  "fst_extract (list_iter (@) [] L) = list_iter (@) [] (map fst_extract L)"
by(induction L, auto)

lemma snd_ext_com_list_it_app:
  "snd_extract (list_iter (@) [] L) = list_iter (@) [] (map snd_extract L)"
by(induction L, auto)

lemma map_com_list_it_app:
  "map F (list_iter (@) [] L) = list_iter (@) [] (map (map F) L)"
by (induction L arbitrary: F, auto)

lemma zip_com_list_it_app:
  "(\<And>L1. length (f L1) = length (g L1)) \<Longrightarrow> zip (list_iter (@) [] (map f L)) (list_iter (@) [] (map g L)) =
    list_iter (@) [] (map (\<lambda>p. zip (f p) (g p)) L)"
by (induction L arbitrary: f g, auto)

lemma list_it_map_app_map:
  "list_iter (@) [] (map (update_snd F) L) = update_snd F (list_iter (@) [] L)"
using fst_ext_com_list_it_app[of L] snd_ext_com_list_it_app[of L]
      map_com_list_it_app[of F "map snd_extract L"]
      zip_com_list_it_app[of "fst_extract" "map F \<circ> snd_extract" L]
by force

lemma update_snd_subset:
  "set L \<subseteq> set L1 \<Longrightarrow> set (update_snd F L) \<subseteq> set (update_snd F L1)"
proof (induction L arbitrary: L1 F)
  case (Cons a L')
    obtain j where H:"j<length L1" "L1 ! j = a"
      using Cons(2)
            set_conv_nth[of L1]
      by auto
    have "(fst a, F (snd a)) \<in> set (update_snd F L1)"
      using H in_set_zip
      by force
    then show ?case
       using Cons
        by simp
qed auto

lemma set_foldl_app[simp]:
  "set(foldl (@) L1 L) = (UN l : set L. set l) \<union> set L1"
by (induction L arbitrary: L1, auto)

lemma set_foldl_union[simp]:
  "foldl (\<union>) S L = (UN l : set L. l) \<union> S"
by (induction L arbitrary: S, auto)

lemma update_snd_rewrite_fun:
  "(\<forall>i<length L. f (snd (L!i)) = g (snd (L!i))) \<Longrightarrow> update_snd f L = update_snd g L"
by (induction L, force+)

lemma update_snd_comp:
  "(update_snd F \<circ> update_snd G) L = update_snd (F \<circ> G) L"
by (induction L, force+)

lemma fst_map:
  "fst_extract L = map fst L"
by (induction L, auto)

lemma snd_map:
  "snd_extract L = map snd L"
by (induction L, auto)

lemma count_rem1:
  "x \<in> set L \<Longrightarrow> count_list L x = count_list (remove1 x L) x + 1"
by(induction L arbitrary: x, auto)

lemma count_Suc:
  "count_list L x = Suc n \<Longrightarrow> x \<in> set L"
proof(induction L arbitrary: n x)
  case (Cons a L1)
    have "x \<notin> set L1 \<Longrightarrow> x = a"
      using count_notin[of x L1]
            Cons(2)
      by (cases "a=x",auto)
    then show ?case
      by (cases "x \<in> set L1", auto)
qed auto

lemma count_inset:
  " x \<in> set L \<Longrightarrow> \<exists>n. count_list L x = Suc n"
proof(induction L arbitrary: x)
  case (Cons a L1)
    then show ?case
      by (cases "x \<in> set L1", auto)
qed auto

lemma same_count_eq:
  "(\<forall>x\<in>set L \<union> set L1. count_list L x = count_list L1 x) \<Longrightarrow> length L1 = length L \<Longrightarrow> (set L1 = set L)"
proof -
  assume count:"\<forall>x\<in>set L \<union> set L1. count_list L x = count_list L1 x"
         and len : "length L1 = length L"
  have 1:"\<And>x. x\<in> set L \<Longrightarrow> x \<in> set L1"
    proof -
      fix x
      assume H: "x\<in>set L"
      have H1: "count_list L x = count_list L1 x"
        using count H
        by auto
      with len H show "x\<in> set L1"
        proof (induction L arbitrary: L1 x)
          case (Cons a L')
            obtain b L1' where H:"L1 = b#L1'"
              using Cons(2) length_Suc_conv
              by metis
            have C: "x = a \<or> x = b \<or> (x\<noteq>a \<and> x \<noteq> b)"
              by auto
            have  " x\<noteq>a \<Longrightarrow> x\<noteq>b \<Longrightarrow> x\<in> set L1"
              using Cons(2-4) H
                    Cons(1)[of L1' x]
              by fastforce
            with C show "x\<in> set L1"
              using Cons(3,4) count_Suc H
              by fastforce+
        qed auto
    qed
  have "\<And>x. x \<in> set L1 \<Longrightarrow> x \<in> set L"
    proof -
      fix x
      assume H: "x\<in>set L1"
      have H1: "count_list L x = count_list L1 x"
        using count H
        by auto
      with len H show "x \<in> set L"
        proof (induction L1 arbitrary: L x)
          case (Cons b L1')
            obtain a L' where H:"L = a#L'"
              using Cons(2) length_Suc_conv
              by metis
            have C: "x = a \<or> x = b \<or> (x\<noteq>a \<and> x \<noteq> b)"
              by auto
            have  " x\<noteq>a \<Longrightarrow> x\<noteq>b \<Longrightarrow> x\<in> set L"
              using Cons(2-4) H
                    Cons(1)[of L' x]
              by fastforce
            with C show "x\<in> set L"
              using Cons(3,4) count_Suc H
              by fastforce+
        qed auto
    qed
  with 1 show ?thesis
    by auto
qed

lemma count_rem1_out:
  "x\<noteq>l1 \<Longrightarrow> count_list (remove1 l1 L) x = count_list L x"
by (induction L arbitrary: l1, auto)

lemma count_conv_length:
  "(\<forall>x\<in>set L. count_list L x = count_list L1 x) \<Longrightarrow> set L \<subseteq> set L1 \<Longrightarrow> length L \<le> length L1"
proof (induction L arbitrary:L1)
  case (Cons l L')
    have A:"\<forall>x\<in>set L'. count_list L' x = count_list (remove1 l L1) x"
      using Cons(2,3) count_rem1[of l L1] count_rem1_out
      by (metis add_diff_cancel_right' count_rem1 remove1.simps(2) set_subset_Cons subsetCE)
    have "set L' \<subseteq> set (remove1 l L1)"
      using Cons(3) A count_notin count_rem1
      by fastforce
    then show ?case
      using Cons(1)[of "remove1 l L1"] A length_remove1[of l L1]
            subsetD[OF Cons(3), of l, simplified]
      by (metis One_nat_def Suc_le_mono Suc_pred length_Cons length_pos_if_in_set)
qed force

lemma same_count_set_length:
  "(\<forall>x\<in>set L. count_list L x = count_list L1 x) \<Longrightarrow> set L = set L1 \<Longrightarrow> length L = length L1"
by (metis order_refl count_conv_length le_antisym)

lemma count_list_app[simp]:
  "count_list (L@L1) x = count_list L x + count_list L1 x"
by (induction L arbitrary: L1, auto)

lemma distinct_fst_imp_count_1:
  "distinct (fst_extract L) \<Longrightarrow> (\<forall>x\<in>set L. count_list L x = 1)"
proof(induction L)
  case (Cons a L')
    thus ?case
      using in_set_zip[of a "fst_extract L'" "snd_extract L'"]
            set_conv_nth[of "fst_extract L'"]
            count_notin set_ConsD set_zip_leftD zip_comp
            length_pos_if_in_set nth_zip
      by (smt One_nat_def add.right_neutral add_Suc_right count_list.simps(2) distinct.simps(2) list_iter.simps(2) prod.collapse)
qed auto

lemma same_count_swap:
  "\<forall>x\<in>set L. count_list L x = count_list L1 x \<Longrightarrow> set L1 \<subseteq> set L \<Longrightarrow> \<forall>x\<in>set L1. count_list L x = count_list L1 x"
proof (rule+)
  fix x
  assume hyp: "\<forall>y\<in>set L. count_list L y = count_list L1 y" "set L1 \<subseteq> set L" "x\<in>set L1"
  thus "count_list L x = count_list L1 x"
    by auto
qed

lemma inset_rem1:
  "x\<in> set L \<Longrightarrow> \<exists>L1 L2. L = L1@[x]@L2 \<and> remove1 x L = L1@L2"
proof(induction L arbitrary: x)
  case (Cons x1 L')
    have "x \<in> set L' \<Longrightarrow> x\<noteq>x1 \<Longrightarrow> ?case"
      proof -
        assume H:"x\<in> set L'" "x\<noteq>x1"
        obtain L1 L2 where H1:"L' = L1 @ [x] @ L2 \<and> remove1 x L' = L1 @ L2"
          using Cons(1)[OF H(1)]
          by auto
        have "x1 # L' = (x1#L1) @ [x] @ L2 \<and> remove1 x (x1 # L') = (x1#L1) @ L2"
         using remove1_append H(2) H1
         by auto
        thus ?case by blast
      qed
    thus ?case using Cons(2) by auto
qed auto

lemma in_set_conv_card_Suc:
  "finite L \<Longrightarrow> x \<in> L \<Longrightarrow>\<exists>k. card L = Suc k"
proof -
  assume H:"x\<in>L" "finite L"
  have 1:"L = insert x (L - {x}) \<and> x \<notin> (L - {x})" using insert_Diff[OF H(1)] by blast
  have "card (L - {x}) = 0 \<longrightarrow> (L - {x}) = {}"
    using card_eq_0_iff[of "L-{x}"] finite_Diff[OF H(2), of "{x}"]
    by meson
  with 1 show "\<exists>k. card L = Suc k"
    using card_Suc_eq[of L "card(L-{x})"]
    by metis
qed

lemma set_zip_subset:
  "set (zip L TL) \<subseteq> set (zip L' TL') \<Longrightarrow> length L' = length TL' \<Longrightarrow> length L = length TL
    \<Longrightarrow> set L \<subseteq> set L' \<and> set TL \<subseteq> set TL'"
proof (induction L arbitrary: TL TL' L')
  case (Cons l L)
    obtain t TL1 where "TL = t#TL1"
      using length_Suc_conv Cons(4)
      by metis
    then show ?case
      using Cons(1)[OF _ Cons(3)] Cons(4,2)
            in_set_zip[of "(l,t)", simplified]
      by auto
qed auto

lemma set_zip_subset_app:
 "length L=length L1 \<Longrightarrow> length L'=length L1' \<Longrightarrow>
        set (zip L L1) \<subseteq> A \<Longrightarrow> set (zip L' L1') \<subseteq> A \<Longrightarrow> set (zip (L@L') (L1@L1')) \<subseteq> A"
proof (induction L arbitrary: L1 L' L1' A)
  case (Cons l' La)
    obtain l1 L1a where "L1 = l1#L1a"
      using Cons(2) length_Suc_conv
      by metis
    then show ?case
      using Cons(1)[OF _ Cons(3) _ Cons(5),of L1a]
            Cons(2,4)
      by simp
qed auto

end